

varstable — Check the stability condition of VAR or SVAR estimates

[Description](#)
[Options](#)
[References](#)

[Quick start](#)
[Remarks and examples](#)
[Also see](#)

[Menu](#)
[Stored results](#)

[Syntax](#)
[Methods and formulas](#)

Description

`varstable` checks the eigenvalue stability condition after estimating the parameters of a vector autoregression using `var` or `svar`.

Quick start

Check eigenvalue stability condition after `var` or `svar`

```
varstable
```

As above, and graph the eigenvalues of the companion matrix

```
varstable, graph
```

As above, and label each eigenvalue with its distance from the unit circle

```
varstable, graph dlabel
```

As above, but label the eigenvalues with their moduli

```
varstable, graph modlabel
```

Menu

Statistics > Multivariate time series > VAR diagnostics and tests > Check stability condition of VAR estimates

Syntax

```
varstable [ , options ]
```

<i>options</i>	Description
Main	
<code>estimates(<i>estname</i>)</code>	use previously stored results <i>estname</i> ; default is to use active results
<code>amat(<i>matrix_name</i>)</code>	save the companion matrix as <i>matrix_name</i>
<code>graph</code>	graph eigenvalues of the companion matrix
<code>dlabel</code>	label eigenvalues with the distance from the unit circle
<code>modlabel</code>	label eigenvalues with the modulus
<code>marker_options</code>	change look of markers (color, size, etc.)
<code>rlopts(<i>cline_options</i>)</code>	affect rendition of reference unit circle
<code>nogrid</code>	suppress polar grid circles
<code>pgrid([...])</code>	specify radii and appearance of polar grid circles; see Options for details
Add plots	
<code>addplot(<i>plot</i>)</code>	add other plots to the generated graph
Y axis, X axis, Titles, Legend, Overall	
<code>twoway_options</code>	any options other than by() documented in [G-3] twoway_options

`varstable` can be used only after `var` or `svar`; see [\[TS\] var](#) and [\[TS\] var svar](#).

Options

Main

`estimates(estname)` requests that `varstable` use the previously obtained set of `var` estimates stored as *estname*. By default, `varstable` uses the active estimation results. See [\[R\] estimates](#) for information on manipulating estimation results.

`amat(matrix_name)` specifies a valid Stata matrix name by which the companion matrix **A** can be saved (see [Methods and formulas](#) for the definition of the matrix **A**). The default is not to save the **A** matrix.

`graph` causes `varstable` to draw a graph of the eigenvalues of the companion matrix.

`dlabel` labels each eigenvalue with its distance from the unit circle. `dlabel` cannot be specified with `modlabel`.

`modlabel` labels the eigenvalues with their moduli. `modlabel` cannot be specified with `dlabel`.

`marker_options` specify the look of markers. This look includes the marker symbol, the marker size, and its color and outline; see [\[G-3\] marker_options](#).

`rlopts(cline_options)` affect the rendition of the reference unit circle; see [\[G-3\] cline_options](#).

`nogrid` suppresses the polar grid circles.

`pgrid([numlist] [, line_options])` determines the radii and appearance of the polar grid circles. By default, the graph includes nine polar grid circles with radii 0.1, 0.2, ..., 0.9 that have the `grid` line style. The *numlist* specifies the radii for the polar grid circles. The *line_options* determine the appearance of the polar grid circles; see [\[G-3\] line_options](#). Because the `pgrid()` option can be repeated, circles with different radii can have distinct appearances.

Add plots

`addplot(plot)` adds specified plots to the generated graph. See [G-3] [addplot_option](#).

Y axis, X axis, Titles, Legend, Overall

`twoway_options` are any of the options documented in [G-3] [twoway_options](#), except by(). These include options for titling the graph (see [G-3] [title_options](#)) and for saving the graph to disk (see [G-3] [saving_option](#)).

Remarks and examples

[stata.com](http://www.stata.com)

Inference after `var` and `svar` requires that variables be covariance stationary. The variables in \mathbf{y}_t are covariance stationary if their first two moments exist and are independent of time. More explicitly, a variable y_t is covariance stationary if

1. $E[y_t]$ is finite and independent of t .
2. $\text{Var}[y_t]$ is finite and independent of t
3. $\text{Cov}[y_t, y_s]$ is a finite function of $|t - s|$ but not of t or s alone.

Interpretation of VAR models, however, requires that an even stricter stability condition be met. If a VAR is stable, it is invertible and has an infinite-order vector moving-average representation. If the VAR is stable, impulse–response functions and forecast-error variance decompositions have known interpretations.

Lütkepohl (2005) and Hamilton (1994) both show that if the modulus of each eigenvalue of the matrix \mathbf{A} is strictly less than one, the estimated VAR is stable (see [Methods and formulas](#) for the definition of the matrix \mathbf{A}).

► Example 1

After fitting a VAR with `var`, we can use `varstable` to check the stability condition. Using the same VAR model that was used in [TS] `var`, we demonstrate the use of `varstable`.


```
. use http://www.stata-press.com/data/r14/lutkepohl2
(Quarterly SA West German macro data, Bil DM, from Lutkepohl 1993 Table E.1)
. var dln_inv dln_inc dln_consump if qtr>=tq(1961q2) & qtr<=tq(1978q4)
(output omitted)
. varstable, graph
Eigenvalue stability condition
```

Eigenvalue	Modulus
.5456253	.545625
-.3785754 + .3853982i	.540232
-.3785754 - .3853982i	.540232
-.0643276 + .4595944i	.464074
-.0643276 - .4595944i	.464074
-.3698058	.369806

All the eigenvalues lie inside the unit circle.
VAR satisfies stability condition.

Because the modulus of each eigenvalue is strictly less than 1, the estimates satisfy the eigenvalue stability condition.

Specifying the `graph` option produced a graph of the eigenvalues with the real components on the x axis and the complex components on the y axis. The graph below indicates visually that these eigenvalues are well inside the unit circle.



◀

► Example 2

This example illustrates two other features of the `varstable` command. First, `varstable` can check the stability of the estimates of the VAR underlying an SVAR fit by `var svar`. Second, `varstable` can check the stability of any previously stored `var` or `var svar` estimates.

We begin by refitting the previous VAR and storing the results as `var1`. Because this is the same VAR that was fit in the [previous example](#), the stability results should be identical.

```
. var dln_inv dln_inc dln_consump if qtr>=tq(1961q2) & qtr<=tq(1978q4)
  (output omitted)
. estimates store var1
```

Now we use `svar` to fit an SVAR with a different underlying VAR and check the estimates of that underlying VAR for stability.

```
. matrix A = (.,0\.,.)
. matrix B = I(2)
. svar d.ln_inc d.ln_consump, aeq(A) beq(B)
  (output omitted)
. varstable
  Eigenvalue stability condition
```

Eigenvalue	Modulus
.548711	.548711
-.2979493 + .4328013i	.525443
-.2979493 - .4328013i	.525443
-.3570825	.357082

All the eigenvalues lie inside the unit circle.
VAR satisfies stability condition.

The `estimates()` option allows us to check the stability of the var results stored as `var1`.

```
. varstable, est(var1)
Eigenvalue stability condition
```

Eigenvalue	Modulus
.5456253	.545625
-.3785754 + .3853982i	.540232
-.3785754 - .3853982i	.540232
-.0643276 + .4595944i	.464074
-.0643276 - .4595944i	.464074
-.3698058	.369806

All the eigenvalues lie inside the unit circle.
VAR satisfies stability condition.

The results are identical to those obtained in the [previous example](#), confirming that we were checking the results in `var1`.

4

Stored results

`varstable` stores the following in `r()`:

Matrices

`r(Re)` real part of the eigenvalues of A
`r(Im)` imaginary part of the eigenvalues of A
`r(Modulus)` modulus of the eigenvalues of A

Methods and formulas

`varstable` forms the companion matrix

$$\mathbf{A} = \begin{pmatrix} \mathbf{A}_1 & \mathbf{A}_2 & \dots & \mathbf{A}_{p-1} & \mathbf{A}_p \\ \mathbf{I} & \mathbf{0} & \dots & \mathbf{0} & \mathbf{0} \\ \mathbf{0} & \mathbf{I} & \dots & \mathbf{0} & \mathbf{0} \\ \vdots & \vdots & \ddots & \vdots & \vdots \\ \mathbf{0} & \mathbf{0} & \dots & \mathbf{I} & \mathbf{0} \end{pmatrix}$$

and obtains its eigenvalues by using `matrix eigenvalues`. The modulus of the complex eigenvalue $r + ci$ is $\sqrt{r^2 + c^2}$. As shown by [Lütkepohl \(2005\)](#) and [Hamilton \(1994\)](#), the VAR is stable if the modulus of each eigenvalue of \mathbf{A} is strictly less than 1.

References

- Hamilton, J. D. 1994. *Time Series Analysis*. Princeton: Princeton University Press.
- Lütkepohl, H. 1993. *Introduction to Multiple Time Series Analysis*. 2nd ed. New York: Springer.
- . 2005. *New Introduction to Multiple Time Series Analysis*. New York: Springer.

Also see

[TS] **var** — Vector autoregressive models

[TS] **var svar** — Structural vector autoregressive models

[TS] **varbasic** — Fit a simple VAR and graph IRFs or FEVDs

[TS] **var intro** — Introduction to vector autoregressive models